

Лекция 14

29 марта

Отказ от указателя фрейма

```
int f(int x, int y) {
    int numerator =
        (x + y) * (x - y);
    int denominator =
        x * x + y * y;
    if (0 == denominator) {
        denominator = 1;
    }
    return (100 * numerator) /
        denominator;
}
```

-fomit-frame-pointer

До 2011 года gcc не включал эту опцию в общие списки оптимизаций

```
f:
    ; пролог
    push    esi
    push    ebx
    mov     esi, dword [esp+12]
    mov     ebx, dword [esp+16]
    ; ...
```

Регистр	Значение
esi	x
ebx	y

Сохраненный регистр	адрес
esi	[esp+4]
ebx	[esp]

Отказ от указателя фрейма

```
int f(int x, int y) {
    int numerator =
        (x + y) * (x - y);
    int denominator =
        x * x + y * y;
    if (0 == denominator) {
        denominator = 1;
    }
    return (100 * numerator) /
        denominator;
}
```

```
f:
    ; ...
    mov     eax, esi
    mov     ecx, ebx
    imul    ecx, ebx ; y * y
    imul    eax, esi ; x * x
    add     ecx, eax ; y * y + x * x
    jne     .L2
    mov     ecx, 1
.L2:
    ; ...
```

Регистр	Значение
esi	x
ebx	y

Отказ от указателя фрейма

```
int f(int x, int y) {
    int numerator =
        (x + y) * (x - y);
    int denominator =
        x * x + y * y;
    if (0 == denominator) {
        denominator = 1;
    }
    return (100 * numerator) /
        denominator;
}
```


```
f:
    ; ...
    lea    eax, [ebx+esi] ; x + y
    sub    esi, ebx      ; x - y
    imul    eax, esi
    imul    eax, eax, 100
    cdq
    idiv    ecx
    ; эпилог
    pop     ebx
    pop     esi
    ret
```

Регистр	Значение
esi	x
ebx	y

x86-64

Регистры x86-64

- Вдвое больше регистров
- Размер регистров удвоился
- Регистры доступны как целиком (64 разряда), так и в виде частей 8, 16, 32 разряда



x86-64

Регистры x86-64: Соглашение по использованию при вызове функций

rax	Возвращаемое значение
rbx	Сохраняется вызванной функцией
rcx	Аргумент #4
rdx	Аргумент #3
rsi	Аргумент #2
rdi	Аргумент #1
rsp	Указатель стека
rbp	Сохраняется вызванной функцией

r8	Аргумент #5
r9	Аргумент #6
r10	Сохраняется вызывающей функцией
r11	Сохраняется вызывающей функцией
r12	Сохраняется вызванной функцией
r13	Сохраняется вызванной функцией
r14	Сохраняется вызванной функцией
r15	Сохраняется вызванной функцией

Регистры x86-64

- Аргументы передаются в функцию через регистры
 - Если целочисленных параметров более 6, остальные передаются через стек
 - Регистры-аргументы могут рассматриваться как сохраненные на стороне вызывающей функции
- Все обращения к фрейму организованы через указатель стека
 - Отпадает необходимость поддерживать значения EBP/RBP
- Остальные регистры
 - 6 регистров сохраняется вызванной функцией
 - 2 регистра сохраняется вызывающей функцией
 - 1 регистр для возвращаемого значения *может рассматриваться как регистр, сохраненный на стороне вызывающей функции*
 - 1 выделенный регистр – указатель стека


x86-64

Обмен значениями переменных long@x86-64

```
void swap_l(long *xp, long *yp) {  
    long t0 = *xp;  
    long t1 = *yp;  
    *xp = t1;  
    *yp = t0;  
}
```

```
swap:  
    mov    rdx,    qword [rdi]  
    mov    rax,    qword [rsi]  
    mov    qword [rdi],    rax  
    mov    qword [rsi],    rdx  
    ret
```

- Параметры передаются через регистры
 - Первый параметр (**xp**) был размещен в **rdi**, второй (**yp**) – в **rsi**
 - 64-разрядные указатели
- Никакие команды не работают со стеком (за исключением **ret**)
- Удалось полностью отказаться от использования стека
 - Все локальные данные размещены на регистрах



x86-64

Локальные переменные в «красной зоне»

Листовая функция

```

/*
 * Обмен через локальный массив
 */
void swap_a(long *xp, long *yp) {
    volatile long loc[2];
    loc[0] = *xp;
    loc[1] = *yp;
    *xp = loc[1];
    *yp = loc[0];
}

```


swap_a:

```

mov rax, qword [rdi]
mov qword [rsp-24], rax
mov rax, qword [rsi]
mov qword [rsp-16], rax
mov rax, qword [rsp-16]
mov qword [rdi], rax
mov rax, qword [rsp-24]
mov qword [rsi], rax
ret

```

- Обходимся без изменения указателя стека
 - Все данные размещены во «фрейме», неявно организованным под текущим указателем стека



x86-64

Нелистовая функция без организации фрейма

```
/* Обмен a[i] и a[i+1] */
void swap_ele(long a[], int i) {
    swap(&a[i], &a[i+1]);
}
```

- На период работы swap уже никаких значений сохранять на регистрах не требуется
- Не требуется сохранять регистры в качестве вызванной функции
- Команда (префикс) `rep` используется вместо команды `NOP`
 - Рекомендации компании AMD для K8

```
swap_ele:
    movsx rsi, esi                ; знаковое расширение i
    lea   rax, [rdi + 8*rsi + 8]  ; &a[i+1]
    lea   rdi, [rdi + 8*rsi]      ; &a[i] первый аргумент
    mov   rsi, rax                ;          второй аргумент
    call  swap
    rep                                ; пустая команда / НОП
    ret
```

x86-64

Пример организации фрейма

```
long sum = 0;
/* Swap a[i] & a[i+1] */
void swap_ele_su
(long a[], int i) {
    swap(&a[i], &a[i+1]);
    sum += (a[i]*a[i+1]);
}
```

- Размещаем значения выражений `&a[i]` и `&a[i+1]` в регистрах, сохраняемых на стороне вызванной функции
- Необходимо сформировать фрейм для сохранения этих регистров

```
swap_ele_su:
    mov     [rsp-16], rbx
    mov     [rsp-8],  rbp
    sub     rsp, 16
    movsx   rax, esi
    lea     rbx, [rdi + 8*rax + 8]
    lea     rbp, [rdi + 8*rax]
    mov     rsi, rbx
    mov     rdi, rbp
    call    swap
    mov     rax, [rbx]
    imul    rax, [rbp]
    add     [rip + sum], rax
    mov     rbx, [rsp]
    mov     rbp, [rsp+8]
    add     rsp, 16
    ret
```

Для x86-64 может использоваться одна из четырех моделей построения кода
-mmodel=[small | medium | large | kernel]

x86-64

Как происходит работа с фреймом

swap_ele_su:

```
mov    [rsp-16], rbx        ; сохраняем rbx
mov    [rsp-8],  rbp        ; сохраняем rbp
sub    rsp, 16              ; выделяем на стеке место для фрейма
movsx  rax, esi             ; знаковое расширение i
lea    rbx, [rdi + 8*rax + 8] ; &a[i+1]
lea    rbp, [rdi + 8*rax]    ; &a[i]
mov    rsi, rbx             ; второй аргумент вызова
mov    rdi, rbp             ; первый аргумент вызова
call   swap
mov    rax, [rbx]           ; помещаем в rax a[i+1]
imul   rax, [rbp]           ; умножаем на a[i]
add    [rip + sum], rax     ; прибавляем к sum
mov    rbx, [rsp]           ; восстанавливаем значение rbx
mov    rbp, [rsp+8]         ; восстанавливаем значение rbp
add    rsp, 16              ; освобождаем место занятое фреймом
ret
```

Особенности работы с фреймом

- Выделение всего фрейма одной командой
 - Обращения к содержимому фрейма используют адресацию относительно `rsp`
 - Уменьшаем значение в указателе стека
 - Выделение памяти может выполняться не сразу, поскольку в определенных временных пределах хранить данные в «красной зоне» безопасно
- Простое освобождение фрейма
 - Увеличиваем значение в указателе стека
 - Указатель фрейма не требуется

A red starburst shape with a black border, containing the text "x86-64" in black.

Промежуточные итоги

x86-64 : организация вызова функций

- Активное использование регистров
 - Передача параметров
 - Больше регистров – больше возможностей вычислять временные значения и их повторно использовать
- Минимальное использование стека
 - Иногда удастся вообще его не использовать
 - Создание/освобождение всего фрейма
- Доступные оптимизации
 - В каком виде будет создан фрейм?
 - Как именно будет выполняться создание?